#-------------------------------------------------
#
# Project created by QtCreator 2013-09-02T21:48:04
#
#-------------------------------------------------

QT       += core xml

TARGET = QtExt
TEMPLATE = lib

CONFIG(debug, debug|release) {
    BUILD_DIR = $$PWD/../../Build/Tools/Debug
    TARGET = $$join(TARGET,,,_d)
} else {
    BUILD_DIR = $$PWD/../../Build/Tools/Release
}
contains(TEMPLATE, "lib") {
    DESTDIR = $$BUILD_DIR/lib
    DLLDESTDIR = $$BUILD_DIR/bin
} else {
    DESTDIR = $$BUILD_DIR/bin
}

UI_DIR = $$BUILD_DIR/build/ui/$$TARGET
OBJECTS_DIR = $$BUILD_DIR/build/obj/$$TARGET
MOC_DIR = $$BUILD_DIR/build/moc/$$TARGET
RCC_DIR = $$BUILD_DIR/build/rcc/$$TARGET


DEFINES += QTEXT_EXPORTS

INCLUDEPATH += ./include


SOURCES += \
    source/QMetaTypeEx.cpp \
    source/QLogger.cpp \
    source/QDocument.cpp \
    source/QDictionary.cpp \
    source/QArray.cpp \
    source/QAlphaBitmap.cpp

HEADERS += \
    include/QtExt.h \
    include/QMetaTypeEx.h \
    include/QLogger.h \
    include/QDocument.h \
    include/QDictionary.h \
    include/QArray.h \
    include/QAlphaBitmap.h

unix:!symbian {
    maemo5 {
        target.path = /opt/usr/lib
    } else {
        target.path = /usr/lib
    }
    INSTALLS += target
}
